



CAPÍTULO 02

Eventos, Comandos, Cursor





Objetivo

Ao final deste capítulo, você será capaz de:

- Identificar quais são os momentos de execução de regras, quais os principais comandos da linguagem LSP e a sintaxe dos cursores, além de diferenciar Cancel(1), Cancel(2) e Cancel(3).

E

ventos, Comandos, Cursor

1.1 Eventos

1.1.1 Eventos do Modelo Gerador

Pré-Seleção: Este evento será chamado ao iniciar a execução de um modelo, logo após a saída da tela de entrada. Nele podem ser tratados ou alterados os valores informados na tela de entrada. Este é o único evento onde pode ser alterado o SELECT que será montado pelo gerador de relatórios, pois é executado antes de efetivar o SELECT no banco e antes de criar as seções que serão executadas, portanto, não é possível listar seções ou manipular controles do relatório neste ponto já que os mesmos não foram criados ainda.

Existem várias funções que podem ser utilizadas para alterar o select, entre elas:

- Função **InsClauSqlWhere** (inserir uma clausula Where no SQL);
- Função **InsClauSQLOrderBy** (incluir uma cláusula de ordenação no SQL);
- Função **DeleteFieldSQL** (retirar campos do SQL);
- Função **SubstituiFrom** (substitui uma clausula FROM no SQL);

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



Abaixo um exemplo que monta cláusulas SQL altera o SELECT a ser executado pelo relatório:

```

Definir Alfa AuxSQL;
Definir Alfa AuxRelac;
Definir Alfa AuxSQLHist;
Definir Alfa xDatRef;
Definir Numero EDatRef;
ConverteDataBanco( EDatRef, xDatRef );
AuxRelac = "";
@ ----- Relacionamento com o Histórico de Filial ( FIXO ) ----- @
@ Não pode utilizar função MontarSQLHistórico, motivo:transferencia de filial
@
AuxSQL = AuxRelac +
      "(R038HFI.DATALT = (SELECT MAX (DATALT) FROM R038HFI TAB2
WHERE "
      + "(TAB2.NUMEMP = R038HFI.NUMEMP) AND "
      + "(TAB2.TIPCOL = R038HFI.TIPCOL) AND "
      + "(TAB2.NUMCAD = R038HFI.NUMCAD) AND "
      + "(TAB2.NUMEMP = TAB2.EMPATU) AND "
      + "(TAB2.NUMCAD = TAB2.CADATU) AND "
      + "(TAB2.DATALT <= " + xDatRef + ")))";
InsClauSQLWhere ( "Detalhe_1", AuxSQL );
AuxRelac = " AND ";
@ ---- Relacionamento com o Historico de Local ( FIXO )---- @
MontarSqlHistorico( "R038HLO", EDatRef, AuxSQLHist );
AuxSQL = AuxRelac + AuxSQLHist;
InsClauSQLWhere( "Detalhe_1", AuxSQL );

```

Seleção: Este evento é utilizado para validar a impressão da seção detalhe Mestre. É Chamado antes de imprimir a seção detalhe principal. Utilizando-se Cancel(1) a seção não será impressa.

A vantagem de sua utilização em relação ao evento "Antes de Imprimir" da seção detalhe é que se nenhum registro for válido, o modelo não será impresso, enquanto na situação anterior seria mostrado uma página em branco.

Exemplo de regra "Seleção" de um relatório:

@ Definir Numero R034FUN.DatAdm Declaracao Criada na Conversao @

PRODUTO	AREA	GESTÃO	CAPITULO		
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor	UNIVERSIDADE CORPORATIVA	

```

Definir Numero xperper; @ Declaracao Criada na Conversao @
Definir Numero xperins; @ Declaracao Criada na Conversao @
Definir Numero EDatRef; @ Declaracao Criada na Conversao @
Definir Cursor Cur_R034CPL;
Definir Data vProExmCpl;
Definir Numero vNumEmp;
Definir Numero vTipCol;
Definir Numero vNumCad;
Se (R034FUN.SitAfa = 7)
    Cancel(1);
vNumEmp=R034FUN.NumEmp;
vTipCol=R034FUN.TipCol;
vNumCad=R034FUN.NumCad;
@----- Busca Dados Complementares -----@
vProExmCpl=0;
Cur_R034CPL.Sql "SELECT PROEXM FROM R034CPL \
                WHERE NUMEMP = :vNumEmp \
                AND TIPCOL = :vTipCol \
                AND NumCad = :vNumCad";
Cur_R034CPL.AbrirCursor();
Se (Cur_R034CPL.Achou)
    Inicio
        vProExmCpl=Cur_R034CPL.ProExm;
        Fim;
Cur_R034CPL.FecharCursor();
RetornaAnoData(DatHoj,anoatu );
RetornaAnoData(R034FUN.DatNas,anonas);
idade = anoatu - anonas;
RetornaMesData(DatHoj,mesatu);
RetornaMesData(R034FUN.DatNas,mesnas);
Se (mesnas > mesatu)
    idade = idade - 1;
xperins = R034FUN.PerIns;
xperper = R034FUN.PerPer;
Se (ETipExa = 6)
    Inicio

```

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

```

Se ((xperins = 0) e (xperper = 0))
    Cancel(2);
Fim;
Se (ETipExa = 12)
    Inicio
    Se ((idade >= 18) e (idade <= 45))
        Cancel(2);
    Fim;
Se ((ETipExa = 12) ou (TipExa = 24))
    Inicio
    Se ((xperins > 0) e (xperper > 0))
        Cancel(2);
    Fim;

```

Inicialização: Este é um evento que será chamado na inicialização da execução do modelo, logo após a saída da tela de entrada mas após o evento “Pré-Seleção”.

Normalmente neste evento inicializa-se variáveis que serão utilizadas na execução do modelo, pois este evento é executado uma única vez na geração do relatório.

Finalização: Esta regra é executada ao término da execução do modelo, ou seja, é a última regra a ser executada pelo modelo. Nela deve-se liberar estruturas alocadas e fechar transações abertas.

Funções Globais: Utilizada para a definição de funções globais que poderão ser chamadas em qualquer ponto do modelo. Nas regras dos relatórios, todas as funções e variáveis são globais (até mesmo as variáveis que são passadas em parâmetros em funções), ou seja, quando criadas estarão visíveis em qualquer outra regra no modelo. A vantagem de criar funções globais é que estas funções podem ser utilizadas várias vezes sem precisar ser redeclaradas ou refeitas, diminuindo e melhorando a manutenção do modelo. Nesta regra é possível calcular valores, listar seções e fazer qualquer outra coisa permitida para as regras em geral.

Abaixo um exemplo de duas funções globais que foram criadas na regra “Funções Globais” para retornar o nome de uma cidade e estado a partir do código e o nome do bairro a partir do código da cidade e do código do bairro.

Conforme dito estas duas funções poderão ser chamadas em qualquer ponto do modelo sempre que for necessário obter tais informações sem precisar definir as funções novamente, apenas passando os parâmetros requeridos pelas mesmas.

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



```

/* FUNCAO QUE RETORNA NOME DA CIDADE E ESTADO */
Definir Funcao RetornaNomCid (numero xcodcid);
Funcao RetornaNomCid (numero xcodcid);
    Inicio
    Definir Cursor CCid;
    Definir Alfa xnomcid;
    Definir Alfa xestcid;
    CCid.Sql "SELECT NOMCID, ESTCID FROM R074CID WHERE CODCID =
:xcodcid";
    CCid.AbrirCursor();
    Se (CCid.Achou)
        Inicio
        xnomcid = CCid.NomCid;
        xestcid = CCid.EstCid;
        Fim;
    Senao
        Inicio
        xnomcid = "";
        xestcid = "";
        Fim;
    CCid.FecharCursor();
    Fim;

/* FUNCAO QUE RETORNA O NOME DO BAIRRO */
Definir Funcao RetornaNomBai(numero xcodcid, numero xcodbai);
Funcao RetornaNomBai (numero xcodcid, numero xcodbai);
    Inicio
    Definir Cursor CBai;
    Definir Alfa xnombai;
    CBai.Sql "SELECT NOMBAI, CEPBAI FROM R074BAI \
        WHERE CODCID = :xcodcid \
        AND CODBAI = :xcodbai";
    CBai.AbrirCursor();
    Se (CBai.Achou)
        Inicio
        xnombai = CBai.NOMBAI;

```

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

```

xcepbai = CBai.CepBai;
Fim;
Senao
    Inicio
    xnombai = "";
    xcepbai = 0;
    Fim;
CBai.FecharCursor();
Fim;

```

Imprimir Página: Esta regra é executada antes de iniciar um nova página, e nela poderão ser setados valores que serão utilizados durante a impressão desta na impressora, como por exemplo , a bandeja de saída. Para alterar a bandeja onde será impressa determinada página, é necessário definir na regra a variável vBandeja do tipo Numero, e atribuir à ela os valores desejados. Após executar esta regra o gerador verifica se existe uma variável com o nome vBandeja e utiliza o valor atribuído à ela para determinar em qual bandeja deve ser impressa a página atual.

```

Definir Numero vBandeja;
Definir Numero xResto;
RestoDivisao(NumPag,2,xResto);
Se (xResto = 1)
    Inicio
    vbandeja = 1;
    Fim;
Senao
    Inicio
    vbandeja = 2;
    Fim;

```

A regra acima determinada que cada página irá imprimir em uma bandeja diferente.

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



1.1.2 Eventos das Seções

Antes de Imprimir: Permite definir regras que irão executar antes da impressão da seção. Este evento é indicado para a implementação de que qualquer operação matemática ou atribuição de valores para campos fórmulas.

Depois de Imprimir: Permite definir uma regra a ser executada depois da impressão da seção.

1.1.3 Eventos dos Controles

Na Impressão: Permite definir uma regra que será executada antes da impressão de qualquer controle inserido no relatório. Este evento está disponível em todos os tipos de relatório (Gráfico, Texto, Arquivo).

Atenção!

Quando se ordena por um campo calculado, este evento SEMPRE deve conter TODA a regra necessária para o valor. Se isto não for feito (toda a regra) o relatório não irá ordenar corretamente por este campo

Este evento NUNCA deve ser utilizado quando uma regra carregar outros campos no relatório, isto porque o sistema não garante a ordem de execução dos controles.



PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



1.2 Comandos

Em razão do número de comandos, será apresentado na apostila apenas uma lista dos comandos utilizado na linguagem LSP, estas palavras são reservadas em qualquer ponte de regra no sistema e em todos as ferramentas de implementação, e por este motivo não podem ser utilizadas como variáveis definidas no sistema.

Resumo dos comandos:

Português	Inglês
Início ou { (abre chaves)	Begin ou { (abre chaves)
Fim ou } (fecha chaves)	End ou } (fecha chaves)
Definir	Declare
Para	For
Enquanto	While
VaPara	Goto
Regra	Rule
Se	If
Senao	Else
Cancel	Cancel
Cancela	Cancel
Alfa	String
Numero	Numeric
Tabela	Array
Funcao	Function
Continue	Continue
Pare	Break
End	Var
E	And
Ou	Or
Abrir	Open
Ler	Read
Fechar	Close
Gravar	Write
Lerln	ReadLn
Formatar	Format

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

Inserir	Include
Grid	Grid
Mensagem	Message
AtualizarCampos	RefreshFields
IniciarTransacao	Transaction
FinalizarTransacao	Commit
DesfazerTransacao	RollBack
ValRet	ValRet
ValStr	ValStr
AbriCursor	OpenCursor
FecharCursor	CloseCursor
Achou	Found
NaoAchou	NotFound
SQL	SQL

Tabela 1 – Comandos

Atenção!

Para visualizar a sintaxe dos comandos apresentados na tabela anterior, acessar o help Ferramentas de Apoio, através do item Sintaxe.



1.3 Cursores

Apesar de ser um item já apresentado no item anterior, apresentaremos mais três exemplos de cursor.

Exemplos:

1) Cursor Retornando apenas uma linha: Quando se tem a chave completa, este cursor deve retornar apenas uma linha, então se usa a seguinte implementação:

Definir Cursor Cur_R076MOT; @Utiliza-se como padrão a nomenclatura Cur_+Nome Tabela@

Definir alfa vmotivo;

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



```

vcodmot = R038HSA.CODMOT;
vmotivo = "";
Cur_R076MOT.Sql "SELECT DESMOT FROM R076MOT WHERE CODMOT =
:vcodmot";
Cur_R076MOT.AbrirCursor();
Se (Cur_R076MOT.Achou)
    Inicio
        @ Encontrou o Registro@
        vmotivo = Cur_R076MOT.DESMOT;
        Fim;
Cur_R076MOT.FecharCursor();

```

2) Cursor Retornando várias linhas: Quando não se tem a chave completa, este cursor pode retornar várias linha, então se usa a seguinte implementação:

Definir Cursor Cur_R038HSA; @Utiliza-se como padrão a nomenclatura Cur_+Nome tabela@

Definir data vperini;

Definir data vperfim;

Vnumemp = R034FUN.NumEmp;

Vtipcol = R034FUN.TipCol;

Vnumcad = R034FUN.NumCad;

MontaData(1,1,1998,vperini);

MontaData(30,01,1998,vperfim);

```

Cur_R038HSA.Sql "SELECT VALSAL FROM R038HSA \
WHERE NUMEMP = :Vnumemp \
AND TIPCOL = :Vtipcol \
AND NUMCAD = :Vnumcad \
AND DATAALT >= :vperini \
AND DATAALT <= :vperfim";

```

Cur_R038HSA.AbrirCursor();

Vvalor = 0;

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



```

Vcontador = 0;
Enquanto (Cur_R038HSA.Achou)
    Inicio
        @ Encontrou o Registro@
        Vvalor = Vvalor + Cur_R038HSA.ValSal;
        Vcontador = Vcontador + 1;
        Cur_R038HSA.Proximo();
    Fim;
Cur_R038HSA.FecharCursor();
Se (Vcontador > 0)
    Formula001 = Vvalor / Vcontador; @ Média dos salários no período @
Senao
    Formula001 = 0;

```

3) Cursor retornando várias linhas, e utilizando a função __Inserir()

para inclusão de cláusulas prontas dentro do cursor:

```

Definir Cursor Cur_R034FUN; @Utiliza-se como padrão a nomenclatura
Cur_+Nome tabela@
Definir Alfa wclausula;
vnumemp = R034FUN.NumEmp;
wclausula = " and (R034FUN.TIPCOL IN (1,2) and (R034FUN.NumCad >= 10
and \
    R034FUN.NumCad <= 100))";

```

```

Cur_R034FUN.Sql "SELECT NUMCAD FROM R034FUN \
    WHERE R034FUN.NUMEMP = :vnumemp __Inserir(:wclausula)";
Cur_R034FUN.AbrirCursor();
Vcontador = 0;
Enquanto (Cur_R034FUN.Achou)
    Inicio
        @ Encontrou o Registro@
        Vcontador = Vcontador + 1;
        Cur_R034FUN.Proximo();
    Fim;
Cur_R034FUN.FecharCursor();
Formula001 = vcontador;

```

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



1.4 Cancel()

Assim como no comando cursor, o comando cancel durante a execução de uma regra pode se comportar de maneira diferente dependendo do evento (lugar) em que está sendo executado. Os comportamentos da função Cancel(1), Cancel(2) e Cancel(3) estão comentados a seguir:

- **Cancel(1)** : Este comando quando utilizado no evento "Na Impressão" de um controle, cancela a execução da regra e a impressão da mesma. Quando é utilizada no evento "Definição\Seleção" de uma seção detalhe, exclui o registro atual do relatório. Se utilizado no evento "Antes de Imprimir" de uma seção, esta apenas terá sua impressão como falsa (apenas oculta a impressão da seção devendo tomar cuidado com as seções subtítulos e subtotais que ficarão visíveis e principalmente com os controles totalizadores destas, pois estes controles irão contabilizar/considerar o registro e/ou valor que estão no detalhe que foi cancelado).
- **Cancel(2)** : Utilizado para imprimir o conteúdo atribuído na variável ValStr em controles do tipo descrição, fórmula, código de barra e memorando, sendo que para isto, a regra deverá ser utilizada no evento Na Impressão dos mesmos. Obs.: Quando este comando é utilizado no evento Antes de Imprimir da seção detalhe, o registro correspondente não será cancelado, pois quando este controle é utilizado nesta propriedade, funciona como se fosse setado a propriedade Imprimir da seção detalhe como Falso, que simplesmente oculta a impressão da seção no modelo. Obs.: Quando utilizado este cancel(2) em controles ou em alguma seção no evento antes de imprimir, os controles totalizadores irão contabilizar/considerar o registro e/ou valor.
- **Cancel(3)** : Utilizado para cancelar a impressão de registros no modelo quando utilizado no Evento Na Impressão dos controles tipo fórmula ou descrição, desde que estes controles estejam setados na ordenação do modelo. Quando não estiver sendo ordenado, apenas cancela a impressão do controle, sendo que neste caso, este controle não será considerado/contabilizado para os controles totalizadores. Este comando quando utilizado nos eventos "Definição\Seleção" e

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



“Definição\Pré-Seleção” do Modelo Gerador, funcionam exatamente igual ao comando Cancel(1), pois os registros também serão cancelados ou o relatório será abortado. Já quando utilizado na propriedade “Antes de Imprimir” de uma seção, excluirá o registro do relatório. Obs.: Quando utilizado este cancel(3) em controles ou em alguma seção no evento antes de imprimir, os controles totalizadores não irão contabilizar/considerar o registro e/ou valor.

1.5 Variáveis Especiais do Gerador de Relatórios

No gerador de relatórios algumas variáveis de sistema possuem um significado especial. Para ter modelos com quebra de organograma é necessário definir na entrada do modelo as variáveis de sistema EspNivTot e EspNivQue, sendo necessário informar um valor para o nível de totais (quebra de local), e o nível de quebra (salto de página). As variáveis especiais possuem nomes e finalidades específicas, entre elas:

- **EspNivTot** – Utilizado como campo de entrada para determinar o nível de totalização do especial. Caso não exista o mesmo definido na entrada, o valor assumido será o do campo especial inserido na seção de detalhe principal.
- **EspNivQue** – Utilizado como campo de entrada para determinar o nível de quebra do especial. Caso não exista o mesmo definido na entrada, o valor assumido será o do campo especial inserido na seção de detalhe principal.

- **EspCodNiv** – Contém o código no nível do especial. Possui no máximo 11 ocorrências;

Exemplo: Para o cadastro de locais: 01 - Empresa Demonstra, 01.02 - Diretoria de finanças, 01.02.01 - Depto contabilidade:

EspCodNiv[1] = retornaria 01;

EspCodNiv[2] = retornaria 02;

- **EspDesNiv** – Contém a descrição do local no nível. Possui no máximo 11 ocorrências;

Exemplo: Para o cadastro de locais: 01 - Empresa Demonstra, 01.02 - Diretoria de finanças, 01.02.01 - Depto contabilidade:

EspDesNiv[1] = retornaria = Empresa Demonstra;

EspDesNiv[2] = retornaria = Diretoria de Finanças;

- **EspNomNiv** – Contém o nome identificador do nível no cadastro do Organograma (ex: 01 - Empresa, 02 - Diretoria, 03 - Depto). Exemplo:

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



RetNomNiv[1] = retornaria Empresa;

RetNomNiv[2] = retornaria Diretoria;

- **EspLevel** – Contém o nível atual em que se está imprimindo;
- **EspTotalLevel** – Contém o nível a ser utilizado pelo totalizador do especial, para efetuar algum cálculo. Esta somente será utilizada em regras, onde existe a necessidade de se somar totalizadores em alguma fórmula.
- **EspMasEdt** – Máscara Editada. Permite definir de que forma será considerada a máscara de edição do local, para relatórios que tenham quebra por organograma. Na tela de entrada deve ter: Nome Variável – EspMasEdt; Descrição (padrão) – Máscara Editada; Tipo – Cadeia de Caracteres; Tamanho – 30; Edição – A[30] e Abrangência – Não.

Exemplo:

1.0.3.4.5.6.7.8.9.A.B – Não executa a quebra do local no nível 2.

1.0.0.0.0.0.0.0.0.0 – Lista apenas a quebra do local no nível 1.

1.6 Comandos para tratamento de Arquivos nas Regras

Dentre os comandos da linguagem LSP, existem alguns comandos que podem ser utilizados no tratamento exclusivo de arquivos texto, permitindo ao usuário operacionalizar este arquivo, que vai desde a criação de um arquivo e suas linhas de registro, até a criação, abertura e/ou gravação de um arquivo texto.

- **Abrir** : Utilizado para abrir um arquivo informado em <nome do arquivo> para o <modo de abertura> informado (Ler/Gravar). Se o arquivo não existir, ele é criado. Ele retorna um manipulador de arquivos.

Sintaxe: Abrir ("<nome do arquivo>",<modo de abertura>);

Exemplo: arq = Abrir ("Teste.txt",Ler);

- **Fechar** : Usado para fecha um arquivo aberto anteriormente pelo comando "Abrir".

Sintaxe: Fechar (<manipulador do arquivo>);

Exemplo: Fechar(arq);

- **Ler** : Lê uma quantidade de caracteres especificados em <tamanho> do arquivo especificado no <manipulador de arquivo> e joga o valor lido na <variável> especificada.

Sintaxe: Ler(<manipulador de arquivo>, <variável>, <tamanho>);

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



Exemplo: Ler(arq,S,20);

- **Gravar** : Grava o valor da <constante> ou da <variável> uma quantidade de caracteres especificados em <tamanho> no arquivo especificado no <manipulador de arquivo>.

Sintaxe: Gravar(<manipulador de arquivo>,<<variável> ou <constante>>,<tamanho>);

Exemplo: Gravar(arq,S,20);

- **Lernl** : Lê uma linha do arquivo indicado pelo <manipulador de arquivo> e joga o valor lido para a <variável> indicada.

Sintaxe: Lernl(<manipulador de arquivo>,<variável>);

Exemplo: Lernl (arq,S);

- **Gravarnl** : Grava uma linha no arquivo indicado pelo <manipulador de arquivo> com o valor da <variável> passada como parâmetro.

Sintaxe: Gravarnl(<manipulador de arquivo>,<<variável> ou <constante>>).

Exemplo: Gravarnl(arq, Str).

1.7 Pesquisar e Selecionar Arquivos na Tela de Entrada

Este recurso permite que o usuário possa pesquisar um determinado arquivo e setar o seu caminho e nome diretamente da pesquisa de arquivos para a entrada.

Para inseri-lo em um modelo é simples. Basta colocar no final do nome da entrada que foi cadastrada a seguinte descrição: "OPENFILE", então quando o modelo for executado irá aparecer um botão que executa a tela de pesquisa de arquivos em diretórios do Windows.




Figura 1 – Valores de Entrada para Tratamento de Arquivos

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



1.8 Funções de Programador

As funções de programador podem ser criadas no Sistema/Módulo ou nas Ferramentas, e são muito utilizadas para otimizar o trabalho do desenvolvedor, pois permitem fazer operações automaticamente nas regras, como por exemplo: retornar valores de cálculos, conversão de dados, montagem de cláusulas SQL, etc...

Devido ao grande número de funções disponíveis no sistema, não é possível ter estas na apostila, devendo os usuários utilizarem sempre que necessário o Help do Sistema ou módulo, ou das Ferramentas (FerramentasApoio.chm).

1.9 Operadores

Os operadores que são utilizados nas regras e fórmulas podem ser:

1.9.1 Operadores Lógicos

Sinal Descrição

= Sinal igual. Utilizado em comparações/operações aritméticas

> Maior que. Utilizado nas comparações

< Menor que. Utilizado nas comparações

< > Diferente de. Utilizado nas comparações

>= Maior ou igual a. Utilizado nas comparações

<= Menor ou igual a. Utilizado nas comparações

e E. Utilizado com o comando "Se", para ligar várias condições, onde todas devem ser verdadeiras para que o resultado da comparação seja verdadeiro.

ou OU. Utilizado com o comando "Se", para ligar várias condições, onde pelo menos uma das condições deve ser verdadeira para que o resultado da comparação seja verdadeiro.

1.9.2 Operadores Aritméticos

Sinal Descrição

= Sinal igual. Utilizado em comparações/operações aritméticas

+ Sinal de Somar. Utilizado nas operações aritméticas de somar

- Sinal de Subtrair. Utilizado operações aritméticas de subtrair

/ Sinal dividir. Utilizado nas operações aritméticas de divisão

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



- * Asterisco. Utilizado nas operações aritméticas de multiplicação
- ++ Incremento de +1. Utilizado para aumentar o valor de uma variável, de um em um.
- Decremento de - 1. Utilizado para diminuir o valor de uma variável, de um em um.

1.9.3 Operadores Extras

Sinal Descrição

- @ Arroba. Delimitador utilizado para incluir comentários que tenham no máximo uma linha.
- /* Início de comentário de bloco de comandos.
- */ Final de comentário de bloco de comandos.

PRODUTO	AREA	GESTÃO	CAPITULO
Tecnologia	Gestão de Pessoas	Gerador de Relatórios Gestão de Pessoas - Avançado	Eventos, Comandos, Cursor

UNIVERSIDADE
CORPORATIVA



